

## Computations with equivariant toric vector bundles

RENÉ BIRKNER, NATHAN OWEN ILTEN, AND LARS PETERSEN

**ABSTRACT.** The study of equivariant vector bundles on a toric variety can yield important geometric information about the underlying variety. The package *ToricVectorBundles* facilitates calculations with such equivariant vector bundles. This package implements two complimentary descriptions of vector bundles and allows for standard operations such as dualizing, direct sums, and tensor, symmetric, and wedge products. Furthermore, *ToricVectorBundles* contains two procedures for calculating the graded cohomology groups of an equivariant vector bundle.

**THE PACKAGE AND ITS APPLICATIONS.** When studying vector bundles on a toric variety, it is natural to first restrict attention to those bundles which are equivariant with respect to the torus action. These equivariant bundles can be described combinatorially in multiple ways: on the one hand, Kaneyama has described equivariant bundles by giving degrees of generators on invariant open subsets along with integer-valued transition matrices, see [Kan]. On the other hand, Klyachko has described equivariant vector bundles by giving vector space filtrations for each ray in the fan corresponding to the underlying toric variety, see [Kly]. Furthermore, many important vector bundles are isomorphic to equivariant ones, including all line bundles as well as the tangent and cotangent bundles for smooth toric varieties.

*ToricVectorBundles* implements the new types `ToricVectorBundleKaneyama` and `ToricVectorBundleKlyachko` in *Macaulay2* [M2], each type corresponding to the respective above-mentioned combinatorial description of equivariant vector bundles. Objects of both types can be created from scratch, either by entering lists of degrees and transition matrices in the first case, or by entering lists of vector space filtrations in the second. For both types, the tangent and cotangent bundles of any smooth toric variety can be automatically generated. Finally, line bundles can easily be generated simply by inputting the coefficients of the prime invariant Weil divisors.

The package contains most of the standard operations for working with vector bundles, including dualizing, direct sums, and tensor, symmetric, and wedge products. Furthermore, the package can compute the cohomology groups of vector bundles in any given degree. For bundles described via the first type, this is done via a graded Čech complex. For bundles described via the second type, this is done via the complex of [Kly, Theorem 4.1.1]. If the underlying variety is complete, *ToricVectorBundles* can calculate a polytope containing all degrees with non-zero cohomology. In this case, the package can then calculate the entire cohomology, regardless of degree.

In *ToricVectorBundles*, every equivariant bundle is stored in two hash tables, whose forms depend on the description being used. For Klyachko’s description, one hash table assigns to each ray of the underlying fan a matrix with the filtration steps, and the other hash table assigns to each ray a basis of the filtered vector space. For Kaneyama’s description, one hash table assigns to each maximal cone

---

2010 *Mathematics Subject Classification.* Primary 14M25; Secondary 14F05, 14Q99, 14-04.  
*ToricVectorBundles* version 1.0.

the degrees of the generator of the bundle over that affine chart, while the other hash table assigns to each cone of codimension one the matrix for the change of bases. All calculations involving convex polyhedral objects are carried out by using the *Macaulay2* package *Polyhedra* [Bir].

The package *ToricVectorBundles* promises to yield applications in a number of areas. In [ES], a description of all possible cohomology tables of vector bundles on projective space is given. The explicit cohomology calculations the package provides may help in finding a proper generalization to describe possible cohomology tables of vector bundles on toric varieties. Another application is in  $(0,2)$  mirror symmetry; here, infinitesimal deformations of the tangent bundle  $\mathcal{T}_X$ , measured by  $H^1(X, \text{End}(\mathcal{T}_X))$ , play an important role, see for example [MM]. Thus, it would be useful to calculate this cohomology group for specific examples. Finally, the cohomology calculations provided by *ToricVectorBundles* may aid in the search for fully strong exceptional sequences of vector bundles on toric varieties, see for example [HP].

EXAMPLES. We will consider the cohomology of equivariant vector bundles on two different toric varieties. We begin with a toric surface:

```
i1 : loadPackage "ToricVectorBundles";
i2 : P1 = convexHull matrix {{1,2,3,3,2,1,0,0},{0,0,1,2,3,3,2,1}};
i3 : F1 = normalFan P1;
```

The fan  $F1$  corresponds to  $\mathbb{P}^1 \times \mathbb{P}^1$  blown up at all four torus fixed points. We now consider the tangent bundle on this surface:

```
i4 : T1 = tangentBundle F1
o4 = {dimension of the variety => 2 }
      number of affine charts => 8
      number of rays => 8
      rank of the vector bundle => 2
o4 : ToricVectorBundleKlyachko
```

From [Ilt, Corollary 1.5], we know that the dimension of the first and second cohomology groups of the tangent bundle must be respectively 4 and 0. We can confirm this:

```
i5 : HH^1(T1)
      4
o5 = (QQ[])
o5 : QQ[]-module, free, degrees {{-1, 0}, {0, -1}, {0, 1}, {1, 0}}
i6 : HH^2(T1)
o6 = 0
o6 : QQ[]-module
```

We now turn our attention to a toric threefold:

```
i7 : P2 = convexHull matrix {
      {1,0,0,-1,0,-1,0,1},
      {0,1,0,-1,0,0,-1,1},
      {0,0,1,0,-1,0,0,0}};
i8 : F2 = faceFan P2;
```

The threefold corresponding to the fan  $F2$  is a smooth Fano threefold of Picard rank 5. In [BB], it was shown that any smooth, complete toric Fano variety is rigid; this was recently generalized to simplicial terminal toric Fanos [dFH]. We confirm the rigidity of our specific threefold:

```

i9 : T2 = tangentBundle F2
o9 = {dimension of the variety => 3 }
      number of affine charts => 12
      number of rays => 8
      rank of the vector bundle => 3
o9 : ToricVectorBundleKlyachko
i10 : HH^1(T2)
o10 = 0
o10 : QQ[]-module

```

However, we shall see that the tangent bundle itself is not rigid. We first consider the cotangent bundle:

```

i11 : Omega = cotangentBundle F2
o11 = {dimension of the variety => 3 }
      number of affine charts => 12
      number of rays => 8
      rank of the vector bundle => 3
o11 : ToricVectorBundleKlyachko

```

Of course, this is just the dual of the tangent bundle:

```

i12 : Omega == dual T2
o12 = true

```

Tensoring these bundles together, we get the endomorphism bundle, whose first cohomology groups measures infinitesimal deformations of the tangent bundle:

```

i13 : Endo = T2 ** Omega
o13 = {dimension of the variety => 3 }
      number of affine charts => 12
      number of rays => 8
      rank of the vector bundle => 9
o13 : ToricVectorBundleKlyachko
i14 : HH^1(Endo)
      29
o14 = (QQ[])
o14 : QQ[]-module, free, degrees {{0, 0, -1}, {0, 0, -1}, {0, 0, -1}, {0, 0, -1}, ...

```

Now, an anticanonical divisor on any toric variety can be given by adding up all invariant prime divisors. Here, we take a canonical divisor using this description:

```

i15 : K = weilToCartier({-1,-1,-1,-1,-1,-1,-1,-1},F2)
o15 = {dimension of the variety => 3 }
      number of affine charts => 12
      number of rays => 8
      rank of the vector bundle => 1
o15 : ToricVectorBundleKlyachko

```

Of course, it should be isomorphic to the third exterior power of the cotangent bundle. We check that this is indeed the case:

```

i16 : areIsomorphic(K,exteriorPower(3,Omega))
o16 = true

```

## REFERENCES.

- [BB] F. Bien and M. Brion, *Automorphisms and local rigidity of regular varieties*, Compositio Math. **104** (1996), no. 1, 1 – 26.
- [Bir] R. Birkner, *Polyhedra: a package for computations with convex polyhedra*, JSAG **1** (2009), 11 – 14.
- [ES] D. Eisenbud and F.-O. Schreyer, *Betti numbers of graded modules and cohomology of vector bundles*, J. Amer. Math. Soc. **22** (2009), no. 3, 859 – 888.
- [dFH] T. de Fernex and C.D. Hacon, *Deformations of canonical pairs and Fano varieties*. available at [arXiv:0901.0389v3](https://arxiv.org/abs/0901.0389v3) [math.AG].
- [HP] L. Hille and M. Perling, *Exceptional Sequences of Invertible Sheaves on Rational Surfaces*. available at [arXiv:0810.1936v2](https://arxiv.org/abs/0810.1936v2) [math.AG].
- [Ilt] N.O. Ilten, *Deformations of Smooth Toric Surfaces*. available at [arXiv:0902.0529v3](https://arxiv.org/abs/0902.0529v3) [math.AG].
- [Kan] T. Kaneyama, *On equivariant vector bundles on an almost homogeneous variety*, Nagoya Math. J. **57** (1975), 65 – 86.
- [Kly] A.A. Klyachko, *Equivariant bundles over toric varieties*, Izv. Akad. Nauk SSSR Ser. Mat. **53** (1989), no. 5, 1001 – 1039, 1135 (Russian); English transl., Math. USSR-Izv. **35** (1990), no. 2, 337 – 375.
- [MM] J. McOrist and I.V. Melnikov, *Summing the instantons in half-twisted linear sigma models*, J. High Energy Phys. **2** (2009), 026, 61.
- [M2] D.R. Grayson and M.E. Stillman, *Macaulay 2, a software system for research in algebraic geometry*, available at [www.math.uiuc.edu/Macaulay2/](http://www.math.uiuc.edu/Macaulay2/).

RECEIVED : 2009-11-26

REVISED : 2010-05-04

ACCEPTED : 2010-06-15

[rbirkner@math.fu-berlin.de](mailto:rbirkner@math.fu-berlin.de) : Mathematisches Institut, Freie Universität Berlin, Arnimallee 3, 14195 Berlin, Germany.

[nilten@cs.uchicago.edu](mailto:nilten@cs.uchicago.edu) : Mathematisches Institut, Freie Universität Berlin, Arnimallee 3, 14195 Berlin, Germany.

[petersen@math.fu-berlin.de](mailto:petersen@math.fu-berlin.de) : Mathematisches Institut, Freie Universität Berlin, Arnimallee 3, 14195 Berlin, Germany.